Example from class. Weight is removed from a piston, resulting in pressure drop from 60 bar to 30 bar. Expansion is irreversible and adiabatic. Compute resulting temperature drop if the initial temperature is 300K. Apply Peng-Robinson equation of state, with temperature-dependent ideal-gas heat capacity.

```
> restart;
> R := 83.14;
```
(cm^3-bar/mol-K)

$$R := 83.14$$

General procedure for Peng-Robinson equation of state. Input are the temperature and volume, along with the material properies: acentric factor (omega), critical temperature, critical pressure. Temperatures should be in Kelvins, pressure in bar, and volume in cm^3/mole

```
> P_PR := proc(T,V,omega,Tc,Pc)
    local Tr,Pr,R,a,b,kappa,alpha;
    Tr := T/Tc;
    Pr := P/Pc;
    R := 83.14;
    a := 0.45724*R^2 * Tc^2 / Pc;
    b := 0.07780 * R * Tc / Pc;
    kappa := 0.37464 + 1.54226*omega - 0.26992 * omega^2;
    alpha := (1 + kappa*(1-sqrt(Tr)))^2;
    evalf(R*T/(V-b) - a * alpha/(V^2 + 2*b*V - b^2));
  end;
```

P_PR := **proc**(*T*, *V*, ω, *Tc*, *Pc*)

local *Tr*, *Pr*, *R*, *a*, *b*, κ, α;

Tr := *T* / *Tc*;

Pr := *P* / *Pc*;

R := 83.14;

a := .45724∗*R*^2∗*Tc*^2 / *Pc*;

b := .07780∗*R*∗*Tc* / *Pc*;

κ := .37464 + 1.54226∗ω – .26992∗ω^2;

α := (1 + κ∗(1 – sqrt(*Tr*)))^2;

evalf(*R*∗*T* / (*V* – *b*) – *a*∗α / (*V*^2 + 2∗*b*∗*V* – *b*^2))

end

```
> P_PR(300,400,.08,250,150);
```
Here's an example of a call to P_PR

56.98987880

Ideal-gas heat capacity (in bar-cm^3/mol-K)

```
> A0 := 38.06; A1 := 1.566; A2 := -8.348e-4; A3 := 1.755e-7;
  cp_ig := T -> A0 + A1*T + A2*T^2 + A3*T^3;
  cv_ig := T -> cp_ig(T) - R;
  cv_approx := cv_ig(300);
```

$$A0 := 38.06$$

$$A1 := 1.566$$

$$A2 := -.0008348$$

$$A3 := .1755\ 10^{-6}$$

$$cp\_ig := T \rightarrow A0 + A1\ T + A2\ T^2 + A3\ T^3$$

$$cv\_ig := T \rightarrow \mathrm{cp\_ig}(T) - R$$

$$cv\_approx := 354.3265000$$

Ideal-gas internal energy (in bar-cm^3/mol), taking U = 0 at T = 300K

```
> U_ig := T -> int(cv_ig(t),t=300..T);
```

$$U\_ig := T \rightarrow \int_{300}^{T} \text{cv\_ig}(t)\, dt$$

Note the care we have to take in the following definitions, distinguishing dummy variables of integration (t, v) from the parameters to the functions (T,V)

```
> dUdV := (T,V,omega,Tc,Pc) ->
  subs(t=T,t*diff(P_PR(t,V,omega,Tc,Pc),t)-P_PR(t,V,omega,Tc,Pc));
```

Here's the internal energy, defined so that it is zero at T = 300, V = infinity (ideal gas state)

```
> U_PR := (T,V,omega,Tc,Pc) -> U_ig(T) +
  int(dUdV(T,v,omega,Tc,Pc),v=infinity..V);
```

$$dUdV := (T, V, \omega, Tc, Pc) \rightarrow \text{subs}\left(t = T, t\left(\frac{\partial}{\partial t}\text{P\_PR}(t, V, \omega, Tc, Pc)\right) - \text{P\_PR}(t, V, \omega, Tc, Pc)\right)$$

$$U\_PR := (T, V, \omega, Tc, Pc) \rightarrow \text{U\_ig}(T) + \int_{\infty}^{V} \text{dUdV}(T, v, \omega, Tc, Pc)\, dv$$

Here's a test evaluation of the derivative dU/dV, and the Peng-Robinson internal energy

```
evalf(dUdV(300,1000,.02,300,200));
evalf(U_PR(300,1000,.02,300,200));
```

1.960934484

-1979.716557

Parameters of problem

```
> P1 := 60;    initial pressure is 60 bar
  P2 := 30;    final pressure is 30 bar
  T1 := 300;   initial temperature is 300 K
```

$$P1 := 60$$

$$P2 := 30$$

$$T1 := 300$$

First we solve it using the ideal-gas law, with a constant heat capacity. For convenience, we assign the equations to be solved to variables

```
> eq1_ig := cv_approx*(T2-T1) = -P2*(V2-V1);
  eq2_ig := P1 = R*T1/V1;
  eq3_ig := P2 = R*T2/V2;
```

$$eq1\_ig := 354.3265000\, T2 - 106297.9500 = -30\, V2 + 30\, V1$$

$$eq2\_ig := 60 = \frac{24942.00}{V1}$$

$$eq3\_ig := 30 = 83.14\,\frac{T2}{V2}$$

Then use those variables in the solve routine

```
> solve({eq1_ig,eq2_ig,eq3_ig},{V1,V2,T2});
```

$$\{V1 = 415.7000000,\ V2 = 752.3966980,\ T2 = 271.4926743\}$$

Here we use the ideal gas model, but with temperature-dependent heat capacity
Four solutions are generated. Make sure you know how to pick the right one.

```
> eq1_ig := U_ig(T2)-U_ig(T1) = -P2*(V2-V1);
  eq2_ig := P1 = R*T1/V1;
  eq3_ig := P2 = R*T2/V2;
> solve({eq1_ig,eq2_ig,eq3_ig},{V1,V2,T2});
```

$$eq1\_ig := -45.08000000\, T2 + .7830000000\, T2^2 - .0002782666667\, T2^3 + .4387500000\, 10^{-7}\, T2^4 - 49788.18750 = -30\, V2 + 30\, V1$$

$$eq2\_ig := 60 = \frac{24942.00}{V1}$$

$$eq3\_ig := 30 = 83.14\,\frac{T2}{V2}$$

$\{ T2 = -290.7109992,\ V1 = 415.7000000,\ V2 = -805.6570825 \},$
$\{ V1 = 415.7000000,\ T2 = 270.3622628,\ V2 = 749.2639510 \},$
$\{ V1 = 415.7000000,\ V2 = 8816.455130 - 7805.879383\ I,\ T2 = 3181.304473 - 2816.651209\ I \},$
$\{ V1 = 415.7000000,\ V2 = 8816.455130 + 7805.879383\ I,\ T2 = 3181.304473 + 2816.651209\ I \}$

Here are the critical properties of ethylene (the gas in the cylinder). They're needed by the Peng-Robinson equation

```
Tc := 282.4;
Pc := 50.4;
omega := 0.089;
```

$$Tc := 282.4$$

$$Pc := 50.4$$

$$\omega := .089$$

Now we set up equations to be solved according to the Peng-Robinson model. Note the spurious generation of imaginary (i.e., involving sqrt(-1)) parts to equation 1. This comes about from the way Maple handles some integrals (generated in setting up U_PR)

```
> eq1 := U_PR(T2,V2,omega,Tc,Pc)-U_PR(T1,V1,omega,Tc,Pc) =
  -P2*(V2-V1);
  eq2 := P1 = P_PR(T1,V1,omega,Tc,Pc);
  eq3 := P2 = P_PR(T2,V2,omega,Tc,Pc);
```

$$eq1 := -45.08000000\ T2 + .7830000000\ T2^2 - .0002782666667\ T2^3 + .4387500000\ 10^{-7}\ T2^4$$
$$- 49788.18750 - .3368776369\ 10^{-7}\ I\ T2 - 121576.5174\ I$$
$$- .2144629645\ 10^{-7}\ \mathrm{arctanh}(\ .01951017789\ V2 + .7071067812\ )\ T2$$
$$- 222403.6700\ \mathrm{arctanh}(\ .01951017789\ V2 + .7071067812\ ) + 7019.223500\ I\ \sqrt{T2}$$
$$+ 4468.576465\ \mathrm{arctanh}(\ .01951017789\ V2 + .7071067812\ )\ \sqrt{T2}$$
$$+ 145005.6552\ \mathrm{arctanh}(\ .01951017789\ V1 + .7071067812\ ) = -30\ V2 + 30\ V1$$

$$eq2 := 60 = \frac{24942.00}{V1 - 36.24296946} - \frac{.4845813566\ 10^{7}}{V1^2 + 72.48593892\ V1 - 1313.552835}$$

$$eq3 := 30 = 83.14\,\frac{T2}{V2 - 36.24296946} - .5001071974\ 10^{7}\,\frac{(\ 1.509763104 - .03033444490\ \sqrt{T2}\ )^2}{V2^2 + 72.48593892\ V2 - 1313.552835}$$

First we try solving with the ideal-gas heat capacity, but the PR equation used for volumetric properties

```
> fsolve({eq1_ig,eq2,eq3},{T2,V2,V1});
```

$\{ V1 = 210.0769490,\ V2 = 524.5156487,\ T2 = 272.1453227 \}$

Now we try solving the full PR equations. Maple returns a complex root. That's not good.

```
> fsolve({eq1,eq2,eq3},{T2,V2,V1});
```

$\{ V1 = 84.69004077 - 62.73151011\ I,\ T2 = 541.4312155 + 446.3474416\ I,$
$V2 = 1525.886253 + 1287.211499\ I \}$

We can try again, giving Maple some idea of an initial starting point. Maple doesn't converge to a solution.

```
> fsolve({eq1,eq2,eq3},{T2,V2,V1},{T2=200..300,V2=200..600,V1=200..5
  00});
Error, (in fsolve/gensys) invalid point generated, try different initial point
```

To simplify things, we can separate the solution for the initial volume V1 from that of the final state points.

```
> v1 := fsolve(eq2,V1);
```

$$v1 := 210.0769490$$

Then try solving for T2 and V2, using the (now known) result for V1 (note that these equations look like what we wrote above, but now we've input v1 (which was just assigned a value) in place of the (previously unknown) V1

```
> eq1 := U_PR(T2,V2,omega,Tc,Pc)-U_PR(T1,v1,omega,Tc,Pc) =
  -P2*(V2-v1);
  eq3 := P2 = P_PR(T2,V2,omega,Tc,Pc);
```

$$eq1 := -45.08000000\,T2 + .7830000000\,T2^2 - .0002782666667\,T2^3 + .4387500000\,10^{-7}\,T2^4 - 19167.62322 - .3368776369\,10^{-7}\,I\,T2 - 349350.8681\,I - .2144629645\,10^{-7}\arctanh(.01951017789\,V2 + .7071067812)\,T2 - 222403.6700\arctanh(.01951017789\,V2 + .7071067812) + 7019.223500\,I\sqrt{T2} + 4468.576465\arctanh(.01951017789\,V2 + .7071067812)\sqrt{T2} = -30\,V2 + 6302.308470$$

$$eq3 := 30 = 83.14\frac{T2}{V2 - 36.24296946} - .5001071974\,10^7\frac{(1.509763104 - .03033444490\sqrt{T2})^2}{V2^2 + 72.48593892\,V2 - 1313.552835}$$

Still, Maple doesn't converge to a solution

```
> fsolve({eq1,eq3},{T2,V2},{T2=220..280,V2=300..550});
Error, (in fsolve/gensys) invalid point generated, try different initial point
```

The problem seems to be from Maple's spurious generation of complex numbers. Look at the equation 1 when we input values for V2 and T2. Probably due to roundoff, the left-hand side is complex

```
> evalf(subs(V2=360,T2=260,eq1));
```

$$-2205.772972 - .0002\,I = -4497.691530$$

We can eliminate this problem by using the "Re" function, which returns the real part of its argument

```
> eq1 := Re(U_PR(T2,V2,omega,Tc,Pc)-U_PR(T1,v1,omega,Tc,Pc)) =
  -P2*(V2-v1);
  eq3 := P2 = P_PR(T2,V2,omega,Tc,Pc);
```

$$eq1 := -19167.62322 + \Re(-45.08000000\,T2 + .7830000000\,T2^2 - .0002782666667\,T2^3 + .4387500000\,10^{-7}\,T2^4 - .3368776369\,10^{-7}\,I\,T2 - .2144629645\,10^{-7}\arctanh(.01951017789\,V2 + .7071067812)\,T2 - 222403.6700\arctanh(.01951017789\,V2 + .7071067812) + 7019.223500\,I\sqrt{T2} + 4468.576465\arctanh(.01951017789\,V2 + .7071067812)\sqrt{T2}) = -30\,V2 + 6302.308470$$

$$eq3 := 30 = 83.14\frac{T2}{V2 - 36.24296946} - .5001071974\,10^7\frac{(1.509763104 - .03033444490\sqrt{T2})^2}{V2^2 + 72.48593892\,V2 - 1313.552835}$$

See:

```
> evalf(subs(V2=360,T2=260,eq1));
```

$$-2205.77297 = -4497.691530$$

Now let's look for the solution again

```
> fsolve({eq1,eq3},{T2,V2},{T2=220..280,V2=300..550});
```

$$\{V2 = 367.2614244,\ T2 = 251.0828899\}$$

Converged! Let's check.

```
> evalf(subs(V2=367.2614244,T2=251.08289,eq1));
```

$$-4715.53422 = -4715.534260$$

```
> evalf(subs(V2=367.2614244,T2=251.08289,eq3));
```

30. = 30.00000005

Looks good. Take another look at how the result varies from the ideal gas model to the Peng-Robinson model.